
The name VHDL is representative of the language itself: it is a two-level acronym
that stands for VHSIC Hardware Description Language; VHSIC stands for very high
speed integrated circuit. The language is vast, verbose, and was originally designed
for modeling digital systems for simulation. As a result, the full definition of the lan-
guage [1] is much larger than what we are concerned with here because many con-
structs in the language (e.g., variables, arbitrary events, floating-point types, delays)
do not have hardware equivalents and hence not synthesizable.

Instead, we focus here on a particular dialect of VHDL dictated in part by the
IEEE standard defining RTL synthesis [2]. Even within this standard, there are many
equivalent ways to do essentially the same thing (e.g., define a process representing
edge-sensitive logic). This document presents a particular idiom that works; it does
not try to define all possible synthesizable VHDL specifications.

1 Structure

Much like a C program is mainly a series of function definitions, a VHDL specification
is mainly a series of entity/architecture definition pairs. An entity is an object with a
series of input and output ports that represent wires or busses, and an architecture
is the “guts” of an entity, comprising concurrent assignment statements, processes,
or instantiations of other entities.

Concurrent assignment statements that use logical expressions to define the val-
ues of signals are one of the most common things in architectures. VHDL supports
the logical operators and, or, nand, nor, xnor, xnor, and not.

library ieee; −− add this to the IEEE library
use ieee.std_logic_1164.all; −− includes std_logic

entity full_adder is
port(a, b, c : in std_logic;

sum, carry : out std_logic);
end full_adder;

architecture imp of full_adder is
begin

sum <= (a xor b) xor c; −− combinational logic
carry <= (a and b) or (a and c) or (b and c);

end imp;

1.1 Components

Once you have defined an entity, the next thing is to instantiate it as a component
within another entity’s architecture.

The interface of the component must be defined in any architecture that instan-
tiates it. Then, any number of port map statements create instances of that compo-
nent.

Here is how to connect two of the full adders to give a two-bit adder:

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all; −− includes unsigned

entity add2 is
port (

A, B : in unsigned(1 downto 0);
C : out unsigned(2 downto 0));

end add2;

architecture imp of add2 is
component full_adder

port (
a, b, c : in std_logic;
sum, carry : out std_logic);

end component;

signal carry : std_logic;

begin

bit0 : full_adder port map (
a => A(0),
b => B(0),
c => ’0’,
sum => C(0),
carry => carry);

bit1 : full_adder port map (
a => A(1),
b => B(1),
c => carry,
sum => C(1),
carry => C(2));

end imp;

1.2 Multiplexers

The when...else construct is one way to specify a multiplexer.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity multiplexer_4_1 is
port(in0, in1, in2, in3 : in unsigned(15 downto 0);

s : in unsigned( 1 downto 0);
z : out unsigned(15 downto 0));

end multiplexer_4_1;

architecture imp of multiplexer_4_1 is
begin

z <= in0 when s = "00" else
in1 when s = "01" else
in2 when s = "10" else
in3 when s = "11" else
(others => ’X’);

end imp;

The with...select is another way to describe a multiplexer.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity multiplexer_4_1 is
port(in0, in1, in2, in3 : in unsigned(15 downto 0);

s0, s1 : in std_logic;
z : out unsigned(15 downto 0));

end multiplexer_4_1;

architecture usewith of multiplexer_4_1 is
signal sels : unsigned(1 downto 0); −− Local wires

begin
sels <= s1 & s0; −− vector concatenation

with sels select
z <= in0 when "00",

in1 when "01",
in2 when "10",
in3 when "11",
(others => ’X’) when others;

end usewith;

1.3 Decoders

Often, you will want to take a set of bits encoded in one way and represent them in
another. For example, the following one-of-eight decoder takes three bits and uses
them to enable one of eight.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity dec1_8 is
port (

sel : in unsigned(2 downto 0);
res : out unsigned(7 downto 0));

end dec1_8;

architecture imp of dec1_8 is
begin

res <= "00000001" when sel = "000" else
"00000010" when sel = "001" else
"00000100" when sel = "010" else
"00001000" when sel = "011" else
"00010000" when sel = "100" else
"00100000" when sel = "101" else
"01000000" when sel = "110" else
"10000000";

end imp;

1.4 Priority Encoders

A priority encoder returns a binary value that indicates the highest set bit among
many. This implementation says the output when none of the bits are set is a “don’t-
care,” meaning the synthesis system is free to generate any output it wants for this
case.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity priority is
port (

sel : in unsigned(7 downto 0);
code : out unsigned(2 downto 0));

end priority;

architecture imp of priority is
begin

code <= "000" when sel(0) = ’1’ else
"001" when sel(1) = ’1’ else
"010" when sel(2) = ’1’ else
"011" when sel(3) = ’1’ else
"100" when sel(4) = ’1’ else
"101" when sel(5) = ’1’ else
"110" when sel(6) = ’1’ else
"111";

end imp;

1.5 Arithmetic Units

VHDL has extensive support for arithmetic. Here is an unsigned 8-bit adder with
carry in and out. By default VHDL’s + operator returns a result that is the same width
as its arguments, so it is necessary to zero-extend them to get the ninth (carry) bit
out. One way to do this is to convert the arguments to integers, add them, then
convert them back.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity adder is
port (

A, B : in unsigned(7 downto 0);
CI : in std_logic;
SUM : out unsigned(7 downto 0);
CO : out std_logic);

end adder;

architecture imp of adder is
signal tmp : unsigned(8 downto 0);
begin

tmp <= A + B + ("0" & CI); −− promote CI to unsigned
SUM <= tmp(7 downto 0);
CO <= tmp(8);

end imp;

A very primitive ALU might perform either addition or subtraction:

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity alu is
port (

A, B : in unsigned(7 downto 0);
ADD : in std_logic;
RES : out unsigned(7 downto 0));

end alu;

architecture imp of alu is
begin

RES <= A + B when ADD = ’1’ else
A − B;

end imp;

VHDL provides the usual arithmetic comparison operators. Note that signed and
unsigned versions behave differently.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity comparator is
port (

A, B : in unsigned(7 downto 0);
GE : out std_logic);

end comparator;

architecture imp of comparator is
begin

GE <= ’1’ when A >= B else ’0’;
end imp;

1.6 Generate statements

To get an unusual array, say that for a 4-bit ripple-carry adder, use a generate con-
struct, which expands its body into multiple gates when synthesized.

library ieee;
use ieee.std_logic_1164.all;

entity rippleadder is
port (a, b : in unsigned(3 downto 0);

cin : in std_logic;
sum : out unsigned(3 downto 0);
cout : out std_logic);

end rippleadder;

architecture imp of rippleadder is
signal c : unsigned(4 downto 0);

begin
c(0) <= cin;

G1: for m in 0 to 3 generate

sum(m) <= a(m) xor b(m) xor c(m);
c(m+1) <= (a(m) and b(m)) or (b(m) and c(m)) or (a(m) and c(m));

end generate G1;

cout <= c(4);
end imp;

2 State-holding Elements

Although there are many ways to express something that behaves like a flip-flop in
VHDL, this is guaranteed to synthesize as you would like

library ieee;
use ieee.std_logic_1164.all;

entity flipflop is
port (Clk, D : in std_logic;

Q : out std_logic);
end flipflop;

architecture imp of flipflop is
begin

process (Clk) −− Process made sensitive to Clk
begin

if rising_edge(Clk) then
Q <= D;

end if;
end process P1;

end imp;

Often, you want a synchronous reset on the flip-flop.

library ieee;
use ieee.std_logic_1164.all;

entity flipflop_reset is
port (Clk, Reset, D : in std_logic;

Q : out std_logic);
end flipflop_reset;

architecture imp of flipflop_reset is
begin

P1: process (Clk)
begin

if rising_edge(Clk) then
if Reset = ’1’ then

Q <= ’0’;
else

Q <= D;
end if;

end if;
end process P1;

end imp;

2.1 Counters

Counters are often useful for delays, dividing clocks, and many other uses. Here is
code for a four-bit unsigned up counter with a synchronous reset:

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity counter is
port(

Clk, Reset : in std_logic;
Q : out unsigned(3 downto 0));

end counter;

architecture imp of counter is
signal count : unsigned(3 downto 0); −− Need an internal signal
begin

process (Clk)
begin

if rising_edge(Clk) then
if Reset = ’1’ then count <= (others => ’0’);
else count <= count + 1;
end if;

end if;
end process;
Q <= count; −− Copy internal signal to output

end imp;

2.2 Shift Registers

Here is code for an eight-bit shift register with serial in and out.

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity shifter is
port (

Clk : in std_logic;
SI : in std_logic;
SO : out std_logic);

end shifter;

architecture impl of shifter is
signal tmp : unsigned(7 downto 0);

begin
process (Clk)
begin

if rising_edge(Clk) then
for i in 0 to 6 loop −− Static loop, expanded at compile time

tmp(i+1) <= tmp(i);
end loop;
tmp(0) <= SI;

end if;
end process;
SO <= tmp(7);

end impl;

2.3 RAMs

While large amounts of memory should be stored off-chip, small RAMs (say 32 × 4
bits) can be implemented directly. Here’s how:

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity ram_32_4 is
port (

Clk : in std_logic;
EN : in std_logic; −− Enable
WE : in std_logic; −− Write enable
addr : in unsigned(4 downto 0);
di : in unsigned(3 downto 0); −− Data in
do : out unsigned(3 downto 0)); −− Data out

end ram_32_4;

architecture imp of ram_32_4 is
type ram_type is array(31 downto 0) of unsigned(3 downto 0);
signal RAM : ram_type;

begin

process (Clk)
begin

if rising_edge(Clk) then
if en = ’1’ then

if we = ’1’ then
RAM(TO_INTEGER(addr)) <= di; −− Write to RAM
do <= di; −− Write−through

else
do <= RAM(TO_INTEGER(addr)); −− Read from RAM

end if;
end if;

end if;
end process;

end imp;

Occasionally, an initialized ROM is the most natural way to compute a certain
function or store some data. Here is what a synchronous ROM looks like:

library ieee;
use ieee.std_logic_1164.all;
use ieee.numeric_std.all;

entity rom_32_4 is
port (

Clk : in std_logic;
en : in std_logic; −− Read enable
addr : in unsigned(4 downto 0);
data : out unsigned(3 downto 0));

end rom_32_4;

architecture imp of rom_32_4 is
type rom_type is array (31 downto 0) of unsigned(3 downto 0);
constant ROM : rom_type :=

("0001", "0010", "0011", "0100", "0101", "0110", "0111", "1000",
"1001", "1010", "1011", "1100", "1101", "1110", "1111", "0001",
"0010", "0011", "0100", "0101", "0110", "0111", "1000", "1001",
"1010", "1011", "1100", "1101", "1110", "1111", "0000", "0010");

begin

process (Clk)
begin

if rising_edge(Clk) then
if en = ’1’ then

data <= ROM(TO_INTEGER(addr));
end if;

end if;
end process;

end imp;

2.4 Finite-State Machines

Write a finite state machine as an entity containing two processes: a sequential pro-
cess with if statement sensitive to the edge of the clock and a combinational process
sensitive to all the inputs of the machine.

library ieee;
use ieee.std_logic_1164.all;

entity tlc is
port (clk, reset : in std_logic;

cars, short, long : in std_logic;
highway_yellow, highway_red : out std_logic;
farm_yellow, farm_red : out std_logic;
start_timer : out std_logic);

end tlc;

architecture imp of tlc is
type states is (HG, HY, FY, FG);
signal state, next_state : states;

begin

process (clk) −− Sequential process
begin

if rising_edge(clk) then
state <= next_state;

end if;
end process;

process (state, reset, cars, short, long) −− Combinational process
begin

highway_yellow <= ’0’;
highway_red <= ’0’;
farm_yellow <= ’0’;
farm_red <= ’0’;
if reset = ’1’ then

start_timer <= ’1’;
next_state <= HG;

else
case state is

when HG =>
farm_red <= ’1’;
if cars = ’1’ and long = ’1’ then

start_timer <= ’1’; next_state <= HY;
else

start_timer <= ’0’; next_state <= HG;
end if;

when HY =>
highway_yellow <= ’1’;
farm_red <= ’1’;
if short = ’1’ then

start_timer <= ’1’; next_state <= FG;
else

start_timer <= ’0’; next_state <= HY;
end if;

when FG =>
highway_red <= ’1’;
if cars = ’0’ or long = ’1’ then

start_timer <= ’1’; next_state <= FY;
else

start_timer <= ’0’; next_state <= FG;
end if;

when FY =>
highway_red <= ’1’;
farm_yellow <= ’1’;
if short = ’1’ then

start_timer <= ’1’; next_state <= HG;
else

start_timer <= ’0’; next_state <= FY;
end if;

end case;
end if;

end process;

end imp;

